<commit_before>DESCRIPTION = "NVIDIA TensorRT (GPU Inference Engine) for deep learning"
HOMEPAGE = "http://developer.nvidia.com/tensorrt"
LICENSE = "Proprietary"
DEPENDS = "cudnn"

SRC_URI = "http://developer.download.nvidia.com/devzone/devcenter/mobile/jetpack_l4t/006/linux-x64/nv-gie-repo-ubuntu1604-6-rc-cuda8.0_${PV}_arm64.deb \
           file://tensorrt-eula.txt"
SRC_URI[md5sum] = "2df905e46c8c2a5a1773659e717591c6"
SRC_URI[sha256sum] = "4d74122a3e4a56083e236d2a3ea9d860c276de4a7037266a8fbabade9556523a"

COMPATIBLE_MACHINE = "(tegra210)"
PACKAGE_ARCH = "${SOC_FAMILY_PKGARCH}"

LIC_FILES_CHKSUM = "file://tensorrt-eula.txt;md5=6cc7c5300a41f51bf22094638b2721c9"

CUDAPATH ?= "/usr/local/cuda-8.0"

S = "${WORKDIR}/var/nv-gie-repo-6-rc-cuda8.0"
B = "${WORKDIR}/build"

do_configure() {
    ln -sf ${WORKDIR}/tensorrt-eula.txt ${S}/
    dpkg-deb --extract ${S}/libgie1_1.0.0-1+cuda8.0_arm64.deb ${B}
    dpkg-deb --extract ${S}/libgie-dev_1.0.0-1+cuda8.0_arm64.deb ${B}
}
do_populate_lic[depends] += "${PN}:do_configure"

do_compile() {
    chrpath -r "${CUDAPATH}/${baselib}" ${B}/usr/lib/aarch64-linux-gnu/lib*${SOLIBS}
}

do_install() {
    install -d ${D}${includedir}
    cp ${B}/usr/include/aarch64-linux-gnu/*.h ${D}${includedir}
    install -d ${D}${libdir}
    cp -d ${B}/usr/lib/aarch64-linux-gnu/* ${D}${libdir}
}

RDEPENDS_${PN} += "libstdc++"
<commit_msg>tensorrt: add missing dependency on dpkg-native<commit_after>DESCRIPTION = "NVIDIA TensorRT (GPU Inference Engine) for deep learning"
HOMEPAGE = "http://developer.nvidia.com/tensorrt"
LICENSE = "Proprietary"
DEPENDS = "cudnn dpkg-native"

SRC_URI = "http://developer.download.nvidia.com/devzone/devcenter/mobile/jetpack_l4t/006/linux-x64/nv-gie-repo-ubuntu1604-6-rc-cuda8.0_${PV}_arm64.deb \
           file://tensorrt-eula.txt"
SRC_URI[md5sum] = "2df905e46c8c2a5a1773659e717591c6"
SRC_URI[sha256sum] = "4d74122a3e4a56083e236d2a3ea9d860c276de4a7037266a8fbabade9556523a"

COMPATIBLE_MACHINE = "(tegra210)"
PACKAGE_ARCH = "${SOC_FAMILY_PKGARCH}"

LIC_FILES_CHKSUM = "file://tensorrt-eula.txt;md5=6cc7c5300a41f51bf22094638b2721c9"

CUDAPATH ?= "/usr/local/cuda-8.0"

S = "${WORKDIR}/var/nv-gie-repo-6-rc-cuda8.0"
B = "${WORKDIR}/build"

do_configure() {
    ln -sf ${WORKDIR}/tensorrt-eula.txt ${S}/
    dpkg-deb --extract ${S}/libgie1_1.0.0-1+cuda8.0_arm64.deb ${B}
    dpkg-deb --extract ${S}/libgie-dev_1.0.0-1+cuda8.0_arm64.deb ${B}
}
do_populate_lic[depends] += "${PN}:do_configure"

do_compile() {
    chrpath -r "${CUDAPATH}/${baselib}" ${B}/usr/lib/aarch64-linux-gnu/lib*${SOLIBS}
}

do_install() {
    install -d ${D}${includedir}
    cp ${B}/usr/include/aarch64-linux-gnu/*.h ${D}${includedir}
    install -d ${D}${libdir}
    cp -d ${B}/usr/lib/aarch64-linux-gnu/* ${D}${libdir}
}

RDEPENDS_${PN} += "libstdc++"

{'fraction_non_alphanumeric': 0.190050107372942, 'fraction_numerical': 0.09878310665712241, 'mean_word_length': 4.692464358452138, 'pattern_counts': {'":': 0, '<': 3, '<?xml version=': 0, '>': 3, 'https://': 0, 'lorem ipsum': 0, 'www.': 0, 'xml': 0}, 'pii_count': 0, 'substrings_counts': 0, 'word_list_counts': {'cursed_substrings.json': 2, 'profanity_word_list.json': 0, 'sexual_word_list.json': 0, 'zh_pornsignals.json': 0}}
{'dir': 'git-commits-cleaned', 'id': '4998785', 'n_tokens_mistral': 1423, 'n_tokens_neox': 1280, 'n_words': 146}